
unit about;

interface

uses
  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms,
  Dialogs, StdCtrls, ShellApi, ExtCtrls;

type
  TAboutForm = class(TForm)
    lblCaption: TLabel;
    Label2: TLabel;
    Label4: TLabel;
    lblUrl: TLabel;
    Label3: TLabel;
    Button1: TButton;
    lblVersion: TLabel;
    Image1: TImage;
    Label5: TLabel;
    Label6: TLabel;
    procedure lblUrlClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
  private
    { Private declarations }
  public
    { Public declarations }
  end;

var
  AboutForm: TAboutForm;

implementation

uses Main;

{$R *.dfm}

procedure TAboutForm.lblUrlClick(Sender: TObject);
var
  str: string;
begin
  TLabel(Sender).cursor := crAppStart;
  application.processmessages;
  str := TLabel(Sender).caption;
  ShellExecute(0, nil, PChar(str), Nil, Nil, SW_NORMAL);
  TLabel(Sender).cursor := crHandPoint;
end;

procedure TAboutForm.FormCreate(Sender: TObject);
var
  i: integer;
begin
  lblVersion.Caption := GetVersion;
  lblVersion.Left := lblCaption.Left + lblCaption.Width - lblVersion.Width;

  Image1.Picture.Icon.Assign(application.Icon);
  i := DpiScale(32);
  if i >= 48 then i := 48
  else if i > 32 then i := 40
  else i := 32;
  Image1.Width := i;
  Image1.Height := i;
  Image1.Picture.Icon.Handle :=
    LoadImage(hINstance, 'MAINICON', IMAGE_ICON, i,i, LR_SHARED);
  Image1.Left := lblCaption.Left - DpiScale(5) - i;
end;

end.

unit main;

(*******************************************************************************
Application:     PDFTK Builder
Description:     A graphical interface to the Windows version of PDFTK.
Module:          Main
Version:         3.10.0
Date:            13-Jul-2018
Compiler:        Delphi 10.1
Author:          Angus Johnson
                 http://www.angusj.com
Copyright &      © 2005-2018
License:         GNU GPL Ver 2 (see http://www.gnu.org/copyleft/gpl.html)
*******************************************************************************)

interface

uses
  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,
  StdCtrls, IniFiles, ShlObj, ShellApi, ActiveX, ComCtrls, Grids, Menus, AnsiStrings,
  StrUtils, ExtCtrls, zlib, ImgList, Buttons, ImageList;

type
  TMainForm = class(TForm)
    gbAdvanced: TGroupBox;
    gbExit: TGroupBox;
    btnBuild: TButton;
    bClose: TButton;
    OpenDialog1: TOpenDialog;
    SaveDialog1: TSaveDialog;
    cbAllowAll: TCheckBox;
    MainMenu1: TMainMenu;
    File1: TMenuItem;
    Close1: TMenuItem;
    Help1: TMenuItem;
    StatusBar1: TStatusBar;
    PageControl1: TPageControl;
    tsCollate: TTabSheet;
    Label2: TLabel;
    lFilename: TLabel;
    lPages: TLabel;
    bAdd: TButton;
    bDelete: TButton;
    bMoveUp: TButton;
    bMoveDown: TButton;
    StringGrid1: TStringGrid;
    ePages: TEdit;
    tsBackground: TTabSheet;
    btnSrcBackground: TButton;
    btnBackground: TButton;
    lblSource_background: TLabel;
    lblBackground: TLabel;
    About1: TMenuItem;
    Contents1: TMenuItem;
    tsSplit: TTabSheet;
    btnSplitSrc: TButton;
    lblSplitSrc: TLabel;
    btnSort: TButton;
    cbPageOneOnly: TCheckBox;
    GroupBox1: TGroupBox;
    rbBackground: TRadioButton;
    rbStamp: TRadioButton;
    tsRotate: TTabSheet;
    Label8: TLabel;
    eRotPagesFrom: TEdit;
    lblSource_rotate: TLabel;
    btnSrcRotate: TButton;
    GroupBox2: TGroupBox;
    rbRotLeft: TRadioButton;
    rbRotRight: TRadioButton;
    rbRot180: TRadioButton;
    Label6: TLabel;
    eRotPagesTo: TEdit;
    pnlProtection: TPanel;
    cbPrinting: TCheckBox;
    cbDegradedPrinting: TCheckBox;
    cbModifyContents: TCheckBox;
    cbScreenReaders: TCheckBox;
    cbModifyAnnotations: TCheckBox;
    cbFillIn: TCheckBox;
    cbAssembly: TCheckBox;
    cbCopyContents: TCheckBox;
    gbPasswords: TGroupBox;
    Label1: TLabel;
    eOwnerPassword: TEdit;
    Label5: TLabel;
    eUserPassword: TEdit;
    rbPageNumber: TRadioButton;
    ImageList1: TImageList;
    Image1: TImage;
    cbJoinBkgndFirstPage: TCheckBox;
    View1: TMenuItem;
    JoinFiles1: TMenuItem;
    SplitFiles1: TMenuItem;
    BackgroundStampNumber1: TMenuItem;
    RotatePages1: TMenuItem;
    N1: TMenuItem;
    mnuShowDocumentProtection: TMenuItem;
    procedure FormCreate(Sender: TObject);
    procedure bCloseClick(Sender: TObject);
    procedure btnBuildClick(Sender: TObject);
    procedure bAddClick(Sender: TObject);
    procedure FormDestroy(Sender: TObject);
    procedure bMoveUpClick(Sender: TObject);
    procedure bMoveDownClick(Sender: TObject);
    procedure bDeleteClick(Sender: TObject);
    procedure ePagesChange(Sender: TObject);
    procedure cbAllowAllClick(Sender: TObject);
    procedure cbPrintingClick(Sender: TObject);
    procedure PageControl1Change(Sender: TObject);
    procedure btnSrcBackgroundClick(Sender: TObject);
    procedure btnBackgroundClick(Sender: TObject);
    procedure About1Click(Sender: TObject);
    procedure Contents1Click(Sender: TObject);
    procedure btnSplitSrcClick(Sender: TObject);
    procedure btnSortClick(Sender: TObject);
    procedure eRotPagesFromChange(Sender: TObject);
    procedure btnSrcRotateClick(Sender: TObject);
    procedure tsCollateResize(Sender: TObject);
    procedure FormCloseQuery(Sender: TObject; var CanClose: Boolean);
    procedure rbPageNumberClick(Sender: TObject);
    procedure gbAdvancedDblClick(Sender: TObject);
    procedure FormKeyDown(Sender: TObject; var Key: Word; Shift: TShiftState);
    procedure StringGrid1Click(Sender: TObject);
    procedure Image1MouseEnter(Sender: TObject);
    procedure Image1MouseLeave(Sender: TObject);
    procedure RotatePages1Click(Sender: TObject);
  private
    fFileList: TStringList;
    fRotatePageCnt: integer;
    fSgOldWindowProc: TWndMethod;
    fNumPageSize: TSize;          //page size of page number PDF
    fNumOffset: TSize;            //position of page number
    fNumSize: integer;            //text size of page number
    procedure SgNewWindowProc(var Message: TMessage);
    procedure RefreshGrid(RowSelected: integer);
    procedure ClearFileList;
    procedure UpdateButtonsAndMenus(RowSelected: integer);
    function FmtPagesString(PagesStr: string; out IsValid: boolean; const fileMarker: string = ''): string;
    procedure DeleteAllTmpFiles;
    procedure LoadIniFileInfo;
    procedure SaveIniFileInfo;

    procedure DoCollateFiles;
    procedure DoSplitFile;
    procedure DoBackgroundStamp(IsBackgroundOp: boolean);
    procedure DoPageNumber;
    procedure DoRotate;
  protected
    procedure CreateWnd; override;
    procedure DestroyWnd; override;
  end;

  PSourceFileRec = ^TSourceFileRec;
  TSourceFileRec = record
    FullPath: string;
    Filename: string;
    Pages: string;
    PageCount: integer;
  end;

  PColorEntry = ^TColorEntry;
  TColorEntry = packed record
    case Boolean of
      false: (R, G, B, A: Byte);
      true: (ARGB: Cardinal);
  end;

var
  MainForm: TMainForm;

  Path, tmpPath, user: string;
  Origin: TPoint;
  Expanded: Boolean;
  ErrorNum: DWORD;
  allowParams, ownerParam, userParam: string;

  MulTable: array [Byte, Byte] of Byte;

const
  PAGES_NOT_COUNTED = -2;

function DPIScale(value: integer): integer;
function GetVersion(): string;

implementation

uses About;

resourcestring
  s_skipFirstPageNumbering  = 'Skip numbering the firs&t page';
  s_chooseBackground        = '&Choose Background ...';
  s_backgroundFirstPageOnly = 'Apply background to only the firs&t page';
  s_chooseStamp             = '&Choose Stamp ...';
  s_stampFirstPageOnly      = 'Apply stamp to only the firs&t page';
  s_save                    = 'S&ave';
  s_saveAs                  = 'Save &As';
  s_saved                   = 'saved.';
  s_doesNotExistInAppFolder = ' does not exist in the application folder.';
  s_noPagesToNumberError    = 'Error - there are no pages to number';
  s_invalidPageRange        = 'Invalid page range specified (file %d)';
  s_version                 = 'Version';
  s_page                    = 'page';
  s_pages                    = 'pages';
  s_filesSelected           = 'files selected';
  s_opFailed                = 'Operation failed.'#10+'Error number: $';
  s_invalidTempPath         = 'Invalid temp path';

{$R *.dfm}
{$R page.res}

type
  TVS_FIXEDFILEINFO = record
    dwSignature: DWORD ;
    dwStrucVersion: DWORD ;
    dwFileVersionMS: DWORD ;
    dwFileVersionLS: DWORD ;
    dwProductVersionMS: DWORD ;
    dwProductVersionLS: DWORD ;
    dwFileFlagsMask: DWORD ;
    dwFileFlags: DWORD ;
    dwFileOS: DWORD ;
    dwFileType: DWORD ;
    dwFileSubtype: DWORD ;
    dwFileDateMS: DWORD ;
    dwFileDateLS: DWORD ;
  end;

  TVS_VERSION_INFO = packed record
    Length          :WORD;
    wValueLength    :WORD;
    wType           :WORD;
    szKey:array[0..Length('VS_VERSION_INFO')] of WideChar;
    Padding1        :array[0..0] of Word;
    FixedInfo       :TVS_FIXEDFILEINFO;
  end;

//------------------------------------------------------------------------------
//------------------------------------------------------------------------------

function DPIScale(value: integer): integer;
begin
  result := value * Screen.PixelsPerInch div 96;
end;

//------------------------------------------------------------------------------


function Blend(const F, B: Cardinal): Cardinal; //nb: B is fully opaque
var
  fg: TColorEntry ABSOLUTE F;
  bg: TColorEntry ABSOLUTE B;
  res: TColorEntry ABSOLUTE Result;
  Fw, Bw: PByteArray;
begin
  if fg.A = 0 then result := B
  else if fg.A = $FF then result := F
  else
  begin
    Fw := @MulTable[fg.A];
    Bw := @MulTable[not fg.A];
    Res.R := Fw[fg.R] + Bw[bg.R];
    Res.G := Fw[fg.G] + Bw[bg.G];
    Res.B := Fw[fg.B] + Bw[bg.B];
    Res.A := 255;
  end;
end;
//------------------------------------------------------------------------------

procedure BlendBackgroundClr(alphaBmp: TBitmap; bkgndColor: TColor);
var
  x,y: Integer;
  c: PColorEntry;
  cl: TColorEntry ABSOLUTE bkgndColor;
begin
  //precondition: requires a 32bit bitmap
  if alphaBmp.PixelFormat <> pf32bit then exit;
  bkgndColor := ColorToRGB(bkgndColor);
  //swap the Red and Blue channels of the background ...
  bkgndColor := (bkgndColor and $FF shl 16) or
    (bkgndColor and $FF00) or (bkgndColor and $FF0000 shr 16);
  cl.A := 255;
  for y := 0 to alphaBmp.Height -1 do
  begin
    c := alphaBmp.ScanLine[y];
    for x := 0 to alphaBmp.Width -1 do
    begin
      c.ARGB := Blend(c.ARGB, cl.ARGB);
      inc(c);
    end;
  end;
end;
//------------------------------------------------------------------------------

function GetUsername: string;
var
  len: DWORD;
begin
  len := 1024;
  SetLength(result,len);
  if windows.GetUserName(PChar(result),len) then
    SetLength(result,len-1) //returned len includes terminating null char
  else
    result := '';
end;
//------------------------------------------------------------------------------

function GetVersion(): string;
var
  rs: TResourceStream;
  vsvi: TVS_VERSION_INFO;
begin
  result := '';
  rs := TResourceStream.CreateFromID(hInstance, 1, RT_VERSION);
  try
    rs.read(vsvi, sizeof(vsvi));
    if vsvi.wValueLength <> sizeof(vsvi.FixedInfo) then exit;
    with vsvi.FixedInfo do
      result := format('%d.%d.%d', [dwFileVersionMS shr 16,
        dwFileVersionMS and $FFFF, dwFileVersionLS shr 16])
  finally
    rs.Free;
  end;
end;
//------------------------------------------------------------------------------

const
  NEVER_GIVE_UP = 0; //GiveUpTimeOutSecs param constant for WinExecAndWait32()

function WinExecAndWait32(const DosCommand: string;
  ShowWindow, GiveUpTimeOutSecs: Word; out textOutput: string): DWord;
const
  BufferSize = 8192;
var
  StartUpInfo: TStartupInfo;
  ProcessInfo: TProcessInformation;
  OutputReadPipeHdl, OutputWritePipeHdl: THandle;
  SecAttribs: TSecurityAttributes;
  Buffer: PAnsiChar;
  timeCnt, BytesRead, BytesAvailable, WaitResult: DWord;
  PipeCreated, ProcessCreated: boolean;

begin
  //nb: 1. The max command line length for CreateProcess() is 32767 characters.

  Result := DWORD(-1); //ie result when unable to create process

  if GiveUpTimeOutSecs = NEVER_GIVE_UP then
    GiveUpTimeOutSecs := $FFFF; //(ie about 18hrs so not quite never :))

  SecAttribs.nLength := SizeOf(TSecurityAttributes) ;
  SecAttribs.bInheritHandle := true;
  SecAttribs.lpSecurityDescriptor := nil;

  PipeCreated := CreatePipe(OutputReadPipeHdl,
    OutputWritePipeHdl, @SecAttribs, BufferSize);

  StartUpInfo.cb := Sizeof(StartUpInfo);
  StartUpInfo.wShowWindow := ShowWindow;

  Buffer := AllocMem(BufferSize + 1);
  try
    if PipeCreated then
    begin
      fillChar( startUpInfo, sizeof(startUpInfo), 0);
      startUpInfo.dwFlags := STARTF_USESHOWWINDOW or STARTF_USESTDHANDLES;
      startUpInfo.hStdInput := 0;
      startUpInfo.hStdOutput := OutputWritePipeHdl;
      startUpInfo.hStdError := OutputWritePipeHdl;
      ProcessCreated := CreateProcess(nil, PWideChar(dosCommand),
        @SecAttribs, @SecAttribs, true, NORMAL_PRIORITY_CLASS,
        nil, nil, startUpInfo, ProcessInfo);
      textOutput := '';
    end else
    begin
      startUpInfo.dwFlags := STARTF_USESHOWWINDOW;
      ProcessCreated := CreateProcess(nil, PWideChar(dosCommand),
        nil, nil, false, NORMAL_PRIORITY_CLASS or CREATE_NEW_CONSOLE,
        nil, nil, startUpInfo, ProcessInfo);
    end;

    if ProcessCreated then
    begin
      timeCnt := 0;
      repeat
        WaitResult := WaitForSingleObject(ProcessInfo.hProcess, 100) ;

        if PipeCreated then
        begin
          //nb: a full pipe buffer would cause an endless loop ...
          if not PeekNamedPipe(OutputReadPipeHdl,
            nil, 0, nil, @BytesAvailable, nil) then break;
          if (BytesAvailable > 0) then
          begin
            //Interestingly, it appears that if the default pipe buffer is larger
            //that that supplied in CreatePipe(), then the default size is used.
            BytesRead := 0;
            if BytesAvailable > BufferSize then
              ReadFile(OutputReadPipeHdl, Buffer[0], BufferSize, BytesRead, nil)
            else
              ReadFile(OutputReadPipeHdl, Buffer[0], BytesAvailable, BytesRead, nil);
            Buffer[BytesRead]:= #0;
            OemToAnsi(Buffer, Buffer) ;
            textOutput := textOutput + string(Buffer);
          end;
        end;

        Application.ProcessMessages;
        inc(timeCnt, 100);
      until (WaitResult <> WAIT_TIMEOUT) or (timeCnt > GiveUpTimeOutSecs *1000);

      if not GetExitCodeProcess(ProcessInfo.hProcess, Result)
        and (result = STILL_ACTIVE) then
          TerminateProcess(ProcessInfo.hProcess, result);

      CloseHandle( ProcessInfo.hProcess );
      CloseHandle( ProcessInfo.hThread );
    end;

    if PipeCreated then
    begin
      CloseHandle(OutputWritePipeHdl);
      CloseHandle(OutputReadPipeHdl);
    end;
  finally
    FreeMem(Buffer) ;
  end;
end;
//------------------------------------------------------------------------------

type
  PPdfObj = ^TPdfObj;
  TPdfObj = record
    number,
    offset: integer;
    filePtr: pAnsiChar;
    stmObjNum: integer;
  end;

function ReversePngFilter(var buffer: PAnsiChar; row_size, buffer_cnt: integer): integer;
var
  i: integer;
  p, end_buff: PAnsiChar;
  filterType: AnsiChar;

  function paeth(c, b, a: byte): byte;
  var
    p,pa,pb,pc: byte;
  begin
    //a = left, b = above, c = upper left
    p := (a + b - c) and $FF;
    pa := abs(p - a);
    pb := abs(p - b);
    pc := abs(p - c);
    if (pa <= pb) and (pa <= pc) then result := a
    else if pb <= pc then result := b
    else result := c;
  end;

  function Average(a, b: byte): byte;
  begin
    //a = left, b = above
    result := (a + b) shr 1;
  end;

begin
  //result will be new buffer length ...
  result := buffer_cnt;
  p := buffer;
  end_buff := buffer + buffer_cnt;
  while p < end_buff do
  begin
    filterType := p^;
    dec(result);
    dec(end_buff);
    move((p +1)^, p^, end_buff - p);
    case filterType of
      #0:
        ;//no filtering used for this row
      #1:
        begin
          //'sub' filtering
          for i := 1 to row_size -1 do
            (p+i)^ := AnsiChar(chr((ord((p+i)^) + ord((p+i -1)^)) and $FF));
        end;
      #2:
        begin
          //'up' filtering
          if p > buffer then //leave top row alone
            for i := 0 to row_size -1 do
              (p+i)^ := AnsiChar(chr((ord((p+i)^) + ord((p+i - row_size)^)) and $FF));
        end;
      #3:
        begin
          //'average' filtering
          if p = buffer then //ie top row
            for i := 1 to row_size -1 do
              (p+i)^ := AnsiChar(chr((ord((p+i)^) + average(ord((p+i -1)^), 0)) and $FF))
          else
          begin
            p^ := AnsiChar(chr((ord((p)^) + average(0, ord((p - row_size)^))) and $FF));
            for i := 1 to row_size -1 do
              (p+i)^ := AnsiChar(chr((ord((p+i)^) +
                average(ord((p+i -1)^), ord((p+i - row_size)^))) and $FF));
          end;
        end;
      #4:
        begin
          //'paeth' filtering
          if p = buffer then //ie top row
            for i := 1 to row_size -1 do
              (p+i)^ := AnsiChar(chr((ord((p+i)^) + paeth(ord((p+i -1)^), 0, 0)) and $FF))
          else
          begin
            p^ := AnsiChar(chr((ord((p)^) + paeth(0, ord((p - row_size)^), 0)) and $FF));
            for i := 1 to row_size -1 do
              (p+i)^ := AnsiChar(chr((ord((p+i)^) + paeth(ord((p+i -1)^),
                ord((p+i - row_size)^), ord((p+i - row_size -1)^))) and $FF));
          end;
        end;
    end;
    inc(p, row_size);
  end;
  ReallocMem(buffer, result);
end;
//------------------------------------------------------------------------------

function GetPdfPageCount(const filename: string): integer;
var
  ms: TMemoryStream;
  i, j, k, cnt, pagesNum, rootNum, predictor, filterColCnt: integer;
  indexArray: array of integer;
  w1,w2,w3: integer;
  p, pEnd, pSaved, pTmp2, buffer, buffPtr: PAnsiChar;
  PdfObj: PPdfObj;
  PdfObjList: TList;

////////////////////////////////////////////////////////////////////////////////
// Summary of steps taken to parse a PDF doc for its page count :-              
// (Who'd have thought it'd take about 1000 lines of code to do that!?)  
////////////////////////////////////////////////////////////////////////////////

//1.  See if there's a 'Linearization dictionary' for easy parsing.
//    Mostly there isn't so ...
//2.  Locate 'startxref' at end of file
//3.  get 'xref' offset and go to xref table
//4.  depending on version the xref table may or may not be in a compressed
//    stream. If it's in a compressed stream (PDF ver 1.5+) then getting the
//    page number requires a LOT of code which is too convoluted to summarise
//    here. Otherwise it still requires a moderate amount of code ...
//5.  parse the xref table and fill a list with object numbers and offsets
//6.  handle subsections within xref table.
//7.  read 'trailer' section at end of each xref
//8.  store 'Root' object number if found in 'trailer'
//9.  if 'Prev' xref found in 'trailer' - loop back to step 3
//10. locate Root in the object list
//11. locate 'Pages' object from Root
//12. get Count from Pages.

  function GetNumber(out num: integer): boolean;
  var
    tmpStr: AnsiString;
  begin
    tmpStr := '';
    while p^ < #33 do inc(p); //skip leading CR,LF & SPC
    while (p^ in ['0'..'9']) do
    begin
      tmpStr := tmpStr + p^;
      inc(p);
    end;
    result := tmpStr <> '';
    if not result then exit;
    num := strtoint(string(tmpStr));
  end;

  function IsString(const str: AnsiString): boolean;
  var
    len: integer;
  begin
    len := length(str);
    result := CompareMem( p, PAnsiChar(str), len);
    if result then inc(p, len); //02-Feb-09: modified
  end;

  function FindStrInDict(const str: AnsiString): boolean;
  var
    nestLvl: integer;
    str1: AnsiChar;
  begin
    //nb: PDF 'dictionaries' start with '<<' and terminate with '>>'
    result := false;
    nestLvl := 0;
    str1 := str[1];
    while not result do
    begin
      while not (p^ in ['>','<',str1]) do inc(p);
      if (p^ = '<') then
      begin
        if (p+1)^ = '<' then begin inc(nestLvl); inc(p); end;
      end
      else if (p^ = '>') then
      begin
        if (p+1)^ = '>' then
        begin
          dec(nestLvl);
          inc(p);
          if nestLvl <= 0 then exit;
        end
      end else
      begin
        result := (nestLvl < 2) and IsString(str);
        if result then exit;
      end;
      inc(p);
    end;
  end;

  function FindEndOfDict: boolean;
  var
    nestLvl: integer;
  begin
    result := true;
    nestLvl := 1;
    while result do
    begin
      while not (p^ in ['>','<']) do inc(p);
      if (p^ = '<') then
      begin
        if (p+1)^ = '<' then begin inc(nestLvl); inc(p); end;
      end
      else if (p+1)^ = '>' then
      begin
        dec(nestLvl);
        if nestLvl < 0 then
          result := false
        else if nestLvl = 0 then
        begin
          inc(p, 2);
          exit; //found end of Dictionary
        end;
        inc(p); //skips first '>'
      end;
      inc(p);
    end;
    result := false;
  end;

  procedure SkipBlankSpace;
  begin
    while (p < pEnd) and (p^ < #33) do inc(p);
  end;

  function GotoObject(objNum: integer): boolean;
  var
    i: integer;
  begin
    i := 0;
    while i < PdfObjList.Count do
      if PPdfObj(PdfObjList[i]).number = objNum then break else inc(i);
    result :=  (i < PdfObjList.Count) and
      (PPdfObj(PdfObjList[i]).filePtr <> nil);
    if not result then exit;
    p := PPdfObj(PdfObjList[i]).filePtr;
    result := GetNumber(i) and (i = objNum);
  end;

  function IsCompressedObject(objNum: integer): boolean;
  var
    i: integer;
  begin
    i := 0;
    while i < PdfObjList.Count do
      if PPdfObj(PdfObjList[i]).number = objNum then break else inc(i);
    result :=  (i < PdfObjList.Count) and
      (PPdfObj(PdfObjList[i]).filePtr = nil);
  end;

  function GetCompressedObjectBuffer(objNum: integer;
    out buffer: PAnsiChar; out bufSize: integer): boolean;
  var
    i, N, First, objIdx: integer;
  begin
    result := false;
    i := 0;
    while i < PdfObjList.Count do
      if PPdfObj(PdfObjList[i]).number = objNum then break else inc(i);
    if (i = PdfObjList.Count) or
      (PPdfObj(PdfObjList[i]).filePtr <> nil) then exit;
    if not GotoObject(PPdfObj(PdfObjList[i]).stmObjNum) then exit;
    objIdx := PPdfObj(PdfObjList[i]).offset;

    pSaved := p;
    if not FindStrInDict('/Type') then exit;
    SkipBlankSpace;
    if not IsString('/ObjStm') then exit;

    p := pSaved;
    //todo - handle uncompressed streams too
    //assume all streams are compressed (though this is really optional)
    if not FindStrInDict('/Filter') then exit;
    SkipBlankSpace;
    //check that this a compression type that we can handle ...
    //nb: [/FlateDecode] is occasionally encountered with Tracker's PDF software
    if not IsString('/FlateDecode') and not IsString('[/FlateDecode]') then exit;
    p := pSaved;
    if not FindStrInDict('/DecodeParms') or not
      FindStrInDict('/Columns') or not GetNumber(filterColCnt) then
        filterColCnt := 0; //j = column count (bytes per row)
    if filterColCnt > 0 then
    begin
      SkipBlankSpace;
      if not IsString('/Predictor') or not GetNumber(predictor) then
        predictor := 0;
    end;

    p := pSaved;
    if not FindStrInDict('/N') then exit;
    //N = number of compressed objects in stream ...
    if not GetNumber(N) or (objIdx > N) then exit;

    p := pSaved;
    if not FindStrInDict('/First') then exit;
    if not GetNumber(First) then exit; //First = first object in stream

    p := pSaved;
    if not FindStrInDict('/Length') then exit;
    if not GetNumber(k) then exit; //k = length of compressed stream
    if not FindEndOfDict then exit;
    while (p^ <> 's') do inc(p);
    if not IsString('stream') then exit;
    SkipBlankSpace;

    //decompress the stream ...
    buffer := nil;
{$IFDEF UNICODE}
    zlib.ZDecompress(p, k, pointer(buffer), bufSize);
{$ELSE}
    DecompressBuf(p, k, k*3, pointer(buffer), bufSize);
{$ENDIF}
    if (filterColCnt > 0) and (predictor > 9) then
      bufSize := ReversePngFilter(buffer, filterColCnt, bufSize);

    p := buffer;
    for i := 0 to objIdx -1 do
    begin
      if not GetNumber(k) then exit;
      if not GetNumber(k) then exit;
    end;
    if not GetNumber(k) or (k <> objNum) then exit; //check it's the right obj
    if not GetNumber(k) then exit; //k = offset of obj from first obj offset

    if objIdx = N then //ie if the last object in the stream
    begin
      p := buffer + First + k;
      bufSize := buffer + bufSize - p;
    end else
    begin
      if not GetNumber(i) then exit; //skip the next object's number
      if not GetNumber(i) then exit; //i == the next object's offset
      p := buffer + First + k;
      bufSize := (i - k); //ie next object's offset minus this object's offset
    end;
    move(p^, buffer^, bufSize);
    ReallocMem(buffer, bufSize);
    p := buffer; //nb: buffer now contains just the uncompressed object
    result := bufSize > 0;
  end;

  function Find(const str: AnsiString; search_forwards: boolean = true): boolean;
  var
    len: integer;
    c: AnsiChar;
  begin
    result := false;
    len := length(str);
    if search_forwards then
    begin
      c := str[1];
      repeat
        while (p < pEnd)  and (p^ <> c) do inc(p);
        if (p = pEnd) then exit;
        if AnsiStrings.StrLComp(p, PAnsiChar(str), len) = 0 then break;
        inc(p);
      until false;
      inc(p,len);
    end else
    begin
      c := str[len];
      repeat
        while (p > PAnsiChar(ms.Memory))  and (p^ <> c) do dec(p);
        if (p = PAnsiChar(ms.Memory)) then exit;
        if AnsiStrings.StrLComp( (p-len+1), PAnsiChar(str), len) = 0 then break;
        dec(p);
      until false;
      inc(p);
    end;
    result := true;
  end;

  function PAnsiCharToInteger(buffer: PAnsiChar; byteCnt: integer): integer;
  begin
    case byteCnt of
      1: result := ord(buffer^);
      2: result := ord(buffer^) shl 8 + ord((buffer+1)^);
      3: result := ord(buffer^) shl 16 +
        ord((buffer+1)^) shl 8 + ord((buffer+2)^);
      4: result := ord(buffer^) shl 24 +
        ord((buffer+1)^) shl 16 + ord((buffer+2)^) +
        ord((buffer+3)^);
      else result := 0;
    end;
  end;

  function GetLinearizedPageNum(out pageNum: integer): boolean;
  var
    pStart,pStop: PAnsiChar;
  begin
    pageNum := -1;
    result := false;
    pStop := p + 32;
    while (p < pStop) and (p^ <> 'o') do inc(p);
    if AnsiStrings.StrLComp( p, 'obj', 3) <> 0 then exit;
    pStart := p;
    if not FindStrInDict('/Linearized') then exit;
    p := pStart;
    if FindStrInDict('/N ') and GetNumber(pageNum) then result := true;
  end;

begin
  //on error return -1 as page count
  result := -1;
  try
    ms := TMemoryStream.Create;
    PdfObjList := TList.Create;
    screen.Cursor := crHourGlass;
    application.ProcessMessages;
    try
      ms.LoadFromFile(filename);

      p := PAnsiChar(ms.Memory);
      pEnd := PAnsiChar(ms.Memory) + ms.Size;

      //for an easy life let's hope the file has a 'linearization dictionary'
      //at the beginning of the document ...
      if GetLinearizedPageNum(result) then exit;

      //find 'startxref' ignoring '%%EOF' at end of file
      p := pEnd -5;
      if not Find('startxref', false) then exit;

      rootNum := -1; //ie flag as not yet found

      if not GetNumber(k) then exit;  //xref offset ==> k
      p :=  PAnsiChar(ms.Memory) + k;

      if AnsiStrings.StrLComp(p, 'xref', 4) <> 0 then
      begin
        //most probably a cross-reference stream (ie PDF doc ver 1.5+)
        //so LOTS of hard work to do now ...

        if not GetNumber(k) then exit; //stream obj number
        if not GetNumber(k) then exit; //stream obj revision number

        pSaved := p;
        if not FindStrInDict('/Type') then exit;
        SkipBlankSpace;
        if not IsString('/XRef') then exit;

        //todo - check for and manage /Prev too

        p := pSaved;
        if not FindStrInDict('/Root') then exit;
        SkipBlankSpace;
        if not GetNumber(rootNum) then exit;
        p := pSaved;

        //todo - handle uncompressed streams too

        //assume all streams are compressed (though this is really optional)
        if not FindStrInDict('/Filter') then exit;
        SkipBlankSpace;
        //check that this a compression type that we can handle ...
        if not IsString('/FlateDecode') then exit;
        p := pSaved;
        if not FindStrInDict('/DecodeParms') or not
          FindStrInDict('/Columns') or not GetNumber(filterColCnt) then
            filterColCnt := 0; //j = column count (bytes per row)
        if filterColCnt > 0 then
        begin
          SkipBlankSpace;
          if not IsString('/Predictor') or not GetNumber(predictor) then
            predictor := 0; //Filtering used prior to Flate (PNG filter >= 10)
        end;

        //get the stream cross-ref table field sizes ...
        p := pSaved;
        if not FindStrInDict('/W') then exit;
        SkipBlankSpace;
        if p^ <> '[' then exit;
        inc(p);
        if not GetNumber(w1) or (w1 <> 1) or not GetNumber(w2) or
          not GetNumber(w3) then exit;

        //Index [F1 N1, ..., Fn, Nn]. If absent assumes F1 = 0 & N based on size
        //(Fn: first object in table subsection; Nn: number in table subsection)
        indexArray := nil;
        p := pSaved;
        if FindStrInDict('/Index') then
        begin
          SkipBlankSpace;
          if p^ <> '[' then exit;
          inc(p);
          while true do
          begin
            k := length(indexArray);
            SetLength(indexArray, k +2);
            if not GetNumber(indexArray[k]) or
              not GetNumber(indexArray[k+1]) then exit;
            pTmp2 := p;
            if not GetNumber(i) then break;
            p := pTmp2;
          end;
        end;

        p := pSaved;
        if not FindStrInDict('/Length') then exit;
        if not GetNumber(k) then exit; //k = length of compressed stream
        if not FindEndOfDict then exit;
        while (p^ <> 's') do inc(p);
        if not IsString('stream') then exit;
        SkipBlankSpace;

        //decompress the stream ...
        buffer := nil;
        try
{$IFDEF UNICODE}
          zlib.ZDecompress(p, k, pointer(buffer), cnt);
{$ELSE}
          DecompressBuf(p, k, k*3, pointer(buffer), cnt);
{$ENDIF}
          //now de-filter the decompressed output (typically PNG filtering)
          //Filter Columns should match the byte count of /W[X Y Z]
          //The decompressed stream prefiltered size == (X Y Z +1) * entries
          //(ie allowing extra byte at the start of each column for filter type)
          //see also http://www.w3.org/TR/PNG-Filters.html

          if (filterColCnt > 0) and (predictor > 9) then
            cnt := ReversePngFilter(buffer, filterColCnt, cnt);

          //make sure the decompressed & defiltered table is a valid size ...
          if cnt mod (w1 + w2 + w3) <> 0 then exit;

          //if the Index array is empty then use the default values ...
          if length(indexArray) = 0 then
          begin
            setLength(indexArray, 2);
            indexArray[0] := 0;
            indexArray[1] := cnt div (w1 + w2 + w3);
          end;

          buffPtr := buffer;
          //loop through each subsection in the table and populate our
          //object list ...
          for i := 1 to (length(indexArray) div 2) do
          begin
            k := indexArray[i*2 -2]; //k := base object number
            for j := 1 to indexArray[i*2 -1] do
            begin
              case buffPtr^ of
                #0: ;//ignore (free object)

                #1: //uncompressed object
                  begin
                    inc(buffPtr);
                    new(PdfObj);
                    PdfObjList.Add(PdfObj);
                    PdfObj.number := k;
                    PdfObj.stmObjNum := -1;
                    PdfObj.offset := PAnsiCharToInteger(buffPtr, w2);
                    PdfObj.filePtr := PAnsiChar(ms.Memory) + PdfObj.offset;
                    dec(buffPtr, w1);
                  end;
                #2: //compressed object
                  begin
                    inc(buffPtr, w1);
                    new(PdfObj);
                    PdfObjList.Add(PdfObj);
                    PdfObj.number := k;
                    PdfObj.stmObjNum := PAnsiCharToInteger(buffPtr, w2);
                    inc(buffPtr,w2);
                    PdfObj.offset := PAnsiCharToInteger(buffPtr, w3);
                    PdfObj.filePtr := nil;
                    dec(buffPtr, w1+w2);
                  end;
              end;
              inc(buffPtr, w1 + w2 + w3);
              inc(k);
            end;
          end;
        finally
          if assigned(buffer) then FreeMem(buffer);
        end;

        if rootNum < 0 then exit;
        if not GotoObject(rootNum) then exit;

        if not FindStrInDict('/Pages') then exit;
        //get the Pages' object number, go to it and get the page count ...
        if not GetNumber(pagesNum) then exit;
        k := -1;
        if IsCompressedObject(pagesNum) then
        begin
          if not GetCompressedObjectBuffer(pagesNum, buffer, cnt) then exit;
          try
            if not FindStrInDict('/Count') or not GetNumber(k) then exit;
          finally
            FreeMem(buffer);
          end;
        end
        else if not GotoObject(pagesNum) or
          not FindStrInDict('/Count') or not GetNumber(k) then exit;

        //if we get this far the page number has been FOUND!!!
        result := k;
        exit;
      end;

      //OK, most likely a PDF doc ver 1.4 (or earlier) ...

      inc(p,4); //ie skip over 'xref'

      while true do //top of loop  //////////////////////////////
      begin
        //get base object number ==> k
        if not GetNumber(k) then exit;
        //get object count ==> cnt
        if not GetNumber(cnt) then exit;
        //it is possible to have 0 objects in a section
        SkipBlankSpace;

        //add all objects in section to list ...
        for cnt := 0 to cnt-1 do
        begin
          new(PdfObj);
          PdfObjList.Add(PdfObj);
          PdfObj.number := k + cnt;
          if not GetNumber(PdfObj.offset) then exit;
          PdfObj.filePtr := PAnsiChar(ms.Memory) + PdfObj.offset;
          //while each entry SHOULD be exactly 20 bytes, not all PDF document
          //creators seems to adhere to this :( ...
          while not (p^ in [#10,#13]) do inc(p);
          while (p^ in [#10,#13]) do inc(p);
        end;
        //check for and process further subsections ...
        if p^ in ['0'..'9'] then continue;

        // parse 'trailer dictionary' ...
        if not IsString('trailer') then exit;
        pSaved := p;
        // get Root (aka Catalog) ...
        if (rootNum = -1) and FindStrInDict('/Root') then
          if not GetNumber(rootNum) then exit;
        p := pSaved;
        if not FindStrInDict('/Prev') then break; //no more xrefs

        //next xref offset ==> k
        if not GetNumber(k) then exit;
        p :=  PAnsiChar(ms.Memory) + k +4;

      end; //bottom of loop /////////////////////////////////////

      //Make sure we've got Root's object number ...
      if rootNum < 0 then exit;
      if not GotoObject(rootNum) then exit;

      if not FindStrInDict('/Pages') then exit;
      //get Pages object number then goto pagesNum ...
      if not GetNumber(pagesNum) or not GotoObject(pagesNum) then exit;
      if not FindStrInDict('/Count') then exit;
      if not GetNumber(cnt) then exit;
      //occasionally the 'count' value is an indirect object
      if GetNumber(k) and IsString(' R') then
      begin
        if not GotoObject(cnt) then exit;
        if not GetNumber(k) or //skip the generation num
          not IsString(' obj') or
          not GetNumber(cnt) then exit;
      end;
      result := cnt; //FOUND!!!!!!

    finally
      screen.Cursor := crDefault;
      for k := 0 to PdfObjList.Count -1 do
        dispose(PPdfObj(PdfObjList[k]));
      PdfObjList.Free;
      ms.Free;
    end;
  except
    //nb: errors are flagged by returning -1
  end;
end;
//------------------------------------------------------------------------------

{$IFNDEF UNICODE}
function CharInSet(c: Char; const theSet: TSysCharset): Boolean;
begin
  result := c in theSet;
end;
{$ENDIF}
//------------------------------------------------------------------------------

function GetTempDirectory: string;
var
  tempFolder: array[0..MAX_PATH] of char;
begin
  if GetTempPath(MAX_PATH, @tempFolder) = 0 then
    raise Exception.Create('GetTempPath: ' + s_invalidTempPath)
  else
    result := tempFolder;
end;
//------------------------------------------------------------------------------

procedure ExecuteCommand(const command: string);
var
  txtOutput: string;
begin
  screen.Cursor := crHourGlass;
  application.ProcessMessages;
  try
    ErrorNum := WinExecAndWait32(command, SW_HIDE, 0, txtOutput);
    if ErrorNum <> 0 then
      raise Exception.Create(txtOutput);
  finally
    screen.Cursor := crDefault;
  end;
end;
//------------------------------------------------------------------------------

procedure ProcessError(const ErrorMessage: string);
begin
  if (ErrorNum <> 0) then
    application.MessageBox(pchar(s_opFailed +
      inttohex(ErrorNum,8)+#10+ ErrorMessage),
      pchar(application.Title), MB_ICONSTOP)
  else
    application.MessageBox(pchar(ErrorMessage),
      pchar(application.Title), MB_ICONSTOP);
end;
//------------------------------------------------------------------------------
// TMainForm methods
//------------------------------------------------------------------------------

procedure TMainForm.FormCreate(Sender: TObject);
var
  bmp: TBitmap;
  desktopRect: TRect;
  imgSize: integer;
begin
  user := GetUsername;
  if user = '' then user := 'user';

  fFileList := TStringList.Create;
  path := extractFilePath(paramstr(0));
  tmpPath := GetTempDirectory + 'pdftkb\';
  fRotatePageCnt := -1;
  LoadIniFileInfo;

  imgSize := DPIScale(16);
  if imgSize >= 24 then imgSize := 24
  else if imgSize > 16 then imgSize := 20
  else imgSize := 16;
  ImageList1.Width := imgSize;
  ImageList1.Height := imgSize;
  image1.Left := gbAdvanced.ClientWidth - imgSize - (DPIScale(5));

  bmp := TBitmap.Create;
  try
    bmp.PixelFormat := pf32bit;
    bmp.LoadFromResourceName(hInstance, 'ARROW' + inttostr(imgSize));
    BlendBackgroundClr(bmp, Self.Color);
    ImageList1.Add(bmp, nil);
    ImageList1.GetBitmap(0, Image1.Picture.Bitmap);
  finally
    bmp.Free;
  end;

  //fixup layout ...
  statusbar1.SimpleText := '  ' + s_version + ' ' + GetVersion();
  PageControl1.ActivePage := tsCollate;
  StatusBar1.Height := DPIScale(20);
  StringGrid1.DefaultRowHeight := DPIScale(18);
  lblSource_background.Caption := '';
  rbPageNumberClick(nil);

  if not Expanded then
  begin
    Expanded := true;
    gbAdvancedDblClick(nil);
  end;

  //hook the StringGrid message handler to catch WM_DROPFILES ...
  fSgOldWindowProc := StringGrid1.WindowProc;
  StringGrid1.WindowProc := SgNewWindowProc;


  //if possible restore the last closed position...
  //nb: Screen.DesktopRect works across *multiple* screens with coords relative
  //to position of topleft corner of primary monitor (that may be on the right).
  desktopRect := Screen.DesktopRect;
  if (origin.X > desktopRect.Left) and (origin.Y > desktopRect.Top) and
    (origin.X + self.Width < desktopRect.Width - desktopRect.Left) and
    (origin.Y + self.Height < desktopRect.Height - desktopRect.Top) then
  begin
    Position := poDesigned;
    Left := origin.X;
    Top := origin.Y;
  end else
    Position := poScreenCenter;
end;
//------------------------------------------------------------------------------

procedure TMainForm.FormDestroy(Sender: TObject);
begin
  ClearFileList;
  fFileList.Free;
  DeleteAllTmpFiles;
  RemoveDir(tmpPath);
  StringGrid1.WindowProc := fSgOldWindowProc;
end;
//------------------------------------------------------------------------------

procedure TMainForm.CreateWnd;
begin
  inherited;
  DragAcceptFiles(StringGrid1.Handle, True);
end;
//------------------------------------------------------------------------------

procedure TMainForm.DestroyWnd;
begin
  DragAcceptFiles(StringGrid1.Handle, False);
  inherited;
end;
//------------------------------------------------------------------------------

procedure TMainForm.FormKeyDown(Sender: TObject; var Key: Word;
  Shift: TShiftState);
var
  sel: TGridRect;
begin
  if (Key = Ord('P')) and (ssAlt in Shift) then
    gbAdvancedDblClick(nil)
  else if (Key = Ord('A')) and (ssCtrl in Shift) and
    (StringGrid1.RowCount > 1) then
  begin
    sel := StringGrid1.Selection;
    sel.Top := 0;
    sel.Bottom := fFileList.Count -1;
    StringGrid1.Selection := sel;
    StringGrid1Click(nil);
  end;
end;
//------------------------------------------------------------------------------

procedure TMainForm.gbAdvancedDblClick(Sender: TObject);
begin
  Image1.Picture := nil; //fixes image refresh problems in DXE2
  if Expanded then
  begin
    gbAdvanced.Height := DPIScale(30);
    Height := Height - DPIScale(186);
    ImageList1.GetBitmap(0, Image1.Picture.Bitmap);
  end else
  begin
    gbAdvanced.Height := DPIScale(222); //designed height (check RCDATA)
    Height := Height + DPIScale(186);
    ImageList1.GetBitmap(1, Image1.Picture.Bitmap);
  end;
  Expanded := not Expanded;
  mnuShowDocumentProtection.Checked := Expanded;
  gbPasswords.Visible := Expanded;
  pnlProtection.Enabled := Expanded;
  cbAllowAll.Enabled := Expanded;
end;
//------------------------------------------------------------------------------

procedure TMainForm.Image1MouseEnter(Sender: TObject);
begin
  Image1.Picture := nil;
  if Expanded then
    ImageList1.GetBitmap(3, Image1.Picture.Bitmap) else
    ImageList1.GetBitmap(2, Image1.Picture.Bitmap);
end;
//------------------------------------------------------------------------------

procedure TMainForm.Image1MouseLeave(Sender: TObject);
begin
  Image1.Picture := nil;
  if Expanded then
    ImageList1.GetBitmap(1, Image1.Picture.Bitmap) else
    ImageList1.GetBitmap(0, Image1.Picture.Bitmap);
end;
//------------------------------------------------------------------------------

procedure TMainForm.SgNewWindowProc(var Message: TMessage);
var
   i, cnt: integer;
   filename : array [0..MAX_PATH] of char;
  SourceFileRec: PSourceFileRec;
begin
  //handle drag and drop of pdf files onto the StringGrid control ...
  if Message.Msg = WM_DROPFILES then
  begin
    cnt := dragqueryfile(message.WParam, $FFFFFFFF, filename, MAX_PATH);
    for i := 0 to cnt-1 do
    begin
      dragqueryfile (message.WParam, i, filename, MAX_PATH);
      if fileexists(filename) and SameText(ExtractFileExt(filename),'.pdf') then
      begin
        new(SourceFileRec);
        SourceFileRec.FullPath := filename;
        SourceFileRec.Filename := ExtractFilename(filename);
        SourceFileRec.PageCount := PAGES_NOT_COUNTED;
        fFileList.AddObject(SourceFileRec.FullPath, pointer(SourceFileRec));
      end;
    end;
    dragfinish(message.WParam);
    RefreshGrid(StringGrid1.Selection.Top);
    UpdateButtonsAndMenus(StringGrid1.Selection.Top);
    tsCollateResize(nil); //in case the vertical scrollbar appears
    Message.Result := 0;
  end
  else fSgOldWindowProc(Message);
end;
//------------------------------------------------------------------------------

procedure TMainForm.bCloseClick(Sender: TObject);
begin
  close;
end;
//------------------------------------------------------------------------------

procedure TMainForm.bAddClick(Sender: TObject);
var
  i: integer;
  SourceFileRec: PSourceFileRec;
  sl: TStringlist;
begin
  OpenDialog1.Options := OpenDialog1.Options + [ofAllowMultiSelect];
  if not OpenDialog1.Execute then exit;

  //we can't sort OpenDialog1.Files so copy the filenames to a TStringlist ...
  sl := TStringlist.Create;
  try
    sl.Assign(OpenDialog1.Files);
    sl.sort;
    for i := 0 to sl.Count -1 do
    begin
      new(SourceFileRec);
      SourceFileRec.FullPath := sl[i];
      SourceFileRec.Filename := ExtractFilename(sl[i]);
      SourceFileRec.PageCount := PAGES_NOT_COUNTED;
      fFileList.AddObject(SourceFileRec.FullPath, pointer(SourceFileRec));
    end;
  finally
    sl.Free;
  end;
  RefreshGrid(StringGrid1.Selection.Top);
  UpdateButtonsAndMenus(StringGrid1.Selection.Top);
  tsCollateResize(nil); //in case vert scrollbar appears
end;
//------------------------------------------------------------------------------

procedure TMainForm.btnSortClick(Sender: TObject);
begin
  fFileList.Sort;
  RefreshGrid(StringGrid1.Selection.Top);
end;
//------------------------------------------------------------------------------

procedure TMainForm.RefreshGrid(RowSelected: integer);
var
  i: integer;
  s, s2: string;
begin
  with StringGrid1 do
    if fFileList.Count = 0 then
    begin
      RowCount := 1;
      Cells[0,0] := '';
      lFilename.caption := '';
      ePages.Text := '';
    end else
    begin
      RowCount := fFileList.Count;
      for i := 0 to fFileList.Count -1 do
        Cells[0,i] := PSourceFileRec(fFileList.Objects[i]).FullPath;

      if StringGrid1.Selection.Bottom > StringGrid1.Selection.Top then
      begin
        lFilename.caption := Format('(%d %s)',
          [StringGrid1.Selection.Bottom - StringGrid1.Selection.Top + 1,
          s_filesSelected]);
      end else
      with PSourceFileRec(fFileList.Objects[RowSelected])^ do
      begin
        ePages.Text := Pages;
        //get pageCount if not already done so ...
        if PageCount = PAGES_NOT_COUNTED then
          PageCount := GetPdfPageCount(FullPath);

        //truncate long filenames...
        if PageCount < 1 then
          s := format('%s  (? %s)',[Filename, s_pages])
        else if PageCount = 1 then
          s := format('%s  (1 %s)',[Filename, s_page])
        else
          s := format('%s  (%d %s)',[Filename, PageCount, s_pages]);
        s2 := s;
        i := self.Canvas.TextWidth(s);
        while i > lFilename.Width do
        begin
          delete(s,8,1);
          i := self.Canvas.TextWidth(s);
        end;
        if s <> s2 then
        begin
          delete(s,8,1);
          insert('...', s, 8);
        end;
        lFilename.caption := s;
      end;
    end;
end;
//------------------------------------------------------------------------------

procedure TMainForm.UpdateButtonsAndMenus(RowSelected: integer);
begin
  if PageControl1.ActivePage = tsCollate then
  begin
    if fFileList.Count = 0 then
    begin
      btnBuild.Enabled := false;
      bDelete.Enabled := false;
      bMoveUp.Enabled := false;
      bMoveDown.Enabled := false;
      ePages.Enabled := false;
      lPages.Enabled := false;
      ePages.Enabled := false;
    end else
    begin
      btnBuild.Enabled := true;
      bDelete.Enabled := true;
      bMoveUp.Enabled := (RowSelected > 0) and
        (StringGrid1.Selection.Top = StringGrid1.Selection.Bottom);
      bMoveDown.Enabled := (RowSelected < StringGrid1.RowCount -1) and
        (StringGrid1.Selection.Top = StringGrid1.Selection.Bottom);
      ePages.Enabled := (fFileList.Count < 27);
      lPages.Enabled := ePages.Enabled;
    end;
  end
  else if PageControl1.ActivePage = tsSplit then
  begin
    btnBuild.Enabled := fileExists(lblSplitSrc.Caption);
  end
  else if PageControl1.ActivePage = tsBackground then
  begin
    btnBuild.Enabled := fileExists(lblSource_background.Caption) and
      (fileExists(lblBackground.Caption) or rbPageNumber.Checked);
  end;
  statusbar1.SimpleText := '  ' + s_version + ' ' + GetVersion();
end;
//------------------------------------------------------------------------------

procedure TMainForm.PageControl1Change(Sender: TObject);
begin
  UpdateButtonsAndMenus(StringGrid1.Selection.Top);
  if PageControl1.ActivePage = tsSplit then
    btnBuild.Caption := s_save else
    btnBuild.Caption := s_saveAs;
end;
//------------------------------------------------------------------------------

procedure TMainForm.ePagesChange(Sender: TObject);
var
  i: integer;
  isValidPageRange: boolean;
begin
  if not ePages.Focused then exit;
  btnBuild.Enabled := true;
  i := StringGrid1.Selection.Top;
  if (i >= fFileList.Count) then exit;
  PSourceFileRec(fFileList.Objects[i]).Pages := ePages.Text;
  RefreshGrid(i);
  FmtPagesString(ePages.Text, isValidPageRange);
  if isValidPageRange then
    ePages.Font.Color := clBlack else
    ePages.Font.Color := clRed;
end;
//------------------------------------------------------------------------------

//FmtPagesString() tidies up (formats) page ranges ready for pdftk.exe so that
//  each range is separated by a space and finishes with a space
//  eg - Input: "1, 2-3,19, 21 - 28" Output: "1 3-6 19 21-28 "
function TMainForm.FmtPagesString(PagesStr: string;
  out IsValid: boolean; const fileMarker: string = ''): string;
var
  i, len: integer;
  rngeStrt, rngeEnd: string;

  procedure SkipSpaces;
  begin
    while (i <= len) and (PagesStr[i] = ' ') do inc(i);
  end;

  function GetNumber(var num: string): boolean;
  begin
    num := '';
    while (i <= len) and CharInSet(PagesStr[i], ['0'..'9']) do
    begin
      num := num + PagesStr[i];
      inc(i);
    end;
    result := num <> '';
  end;

begin
  IsValid := false;
  result := '';
  len := length(PagesStr);
  i := 1;
  while i <= len do
  begin
    SkipSpaces;
    if (i > len) then break;
    if not GetNumber(rngeStrt) then exit;
    SkipSpaces;
    if (i <= len) and (PagesStr[i] = '-') then
    begin
      inc(i);
      SkipSpaces;
      if not GetNumber(rngeEnd) then exit;
      SkipSpaces;
      result := result + fileMarker + rngeStrt + '-' + rngeEnd + ' ';
    end
    else
      result := result + fileMarker + rngeStrt + ' ';

    if (i > len) then break
    else if CharInSet(PagesStr[i], [',',';',':']) then
    begin
      inc(i);
      continue;
    end else
      exit;
  end;
  if result = '' then result := ' ';
  IsValid := true;
end;
//------------------------------------------------------------------------------

procedure TMainForm.ClearFileList;
var
  i: integer;
begin
  for i := 0 to fFileList.Count -1 do
    dispose(PSourceFileRec(fFileList.Objects[i]));
  fFileList.Clear;
end;
//------------------------------------------------------------------------------

procedure TMainForm.StringGrid1Click(Sender: TObject);
begin
  UpdateButtonsAndMenus(StringGrid1.Selection.Top);
  RefreshGrid(StringGrid1.Selection.Top);
end;
//------------------------------------------------------------------------------

procedure TMainForm.bMoveUpClick(Sender: TObject);
var
  i: integer;
  sel: TGridRect;
  SourceFileRec: PSourceFileRec;
begin
  sel := StringGrid1.Selection;
  i := sel.Top;
  SourceFileRec := pointer(fFileList.Objects[i]);
  fFileList.Delete(i);
  fFileList.InsertObject(i-1,SourceFileRec.FullPath,pointer(SourceFileRec));
  dec(sel.Top);
  dec(sel.Bottom);
  RefreshGrid(i-1);
  StringGrid1.Selection := sel;
  if sel.Top < StringGrid1.TopRow then StringGrid1.TopRow := sel.Top;
  UpdateButtonsAndMenus(i-1);
end;
//------------------------------------------------------------------------------

procedure TMainForm.bMoveDownClick(Sender: TObject);
var
  i: integer;
  sel: TGridRect;
  SourceFileRec: PSourceFileRec;
begin
  sel := StringGrid1.Selection;
  i := sel.Top;
  SourceFileRec := pointer(fFileList.Objects[i]);
  fFileList.Delete(i);
  fFileList.InsertObject(i+1,SourceFileRec.FullPath,pointer(SourceFileRec));
  inc(sel.Top);
  inc(sel.Bottom);
  RefreshGrid(i+1);
  StringGrid1.Selection := sel;
  with StringGrid1 do
    if sel.Bottom  >= TopRow + VisibleRowCount then
      StringGrid1.TopRow := sel.Bottom - VisibleRowCount +1;
  UpdateButtonsAndMenus(i+1);
end;
//------------------------------------------------------------------------------

procedure TMainForm.bDeleteClick(Sender: TObject);
var
  i,j: integer;
begin
  i := StringGrid1.Selection.Top;
  for j := StringGrid1.Selection.Bottom downto i do
  begin
    dispose(PSourceFileRec(fFileList.Objects[j]));
    fFileList.Delete(j);
  end;
  if i >= fFileList.Count then i := fFileList.Count -1;
  RefreshGrid(i);
  UpdateButtonsAndMenus(i);
end;
//------------------------------------------------------------------------------

procedure TMainForm.DeleteAllTmpFiles;
var
  i: integer;
  sr: TSearchRec;
begin
  i := FindFirst(tmpPath+'*.*', faAnyFile, sr);
  while i = 0 do
  begin
    SetFileAttributes(pchar(tmpPath+sr.Name), 0); //remove read-only etc
    DeleteFile(tmpPath+sr.Name);
    i := FindNext(sr);
  end;
  FindClose(sr);
end;
//------------------------------------------------------------------------------

procedure TMainForm.DoCollateFiles;
var
  i: integer;
  c: Char;
  s, fileParams, actionParams: string;
  IsValidPageRanges: boolean;
begin
  ChDir(tmpPath);

  //nb: Since PDFTK has limited 'file handles' (ie A..Z), if there are
  //more than 26 files, we simply collate the files using the '*' wildcard
  //char. (However, using '*' prevents page selection for individual files.)
  if fFileList.Count > 26 then
  begin
    //copy all the pdf files into a temp folder ready for collating
    //while preserving sort order ...
    for i := 0 to fFileList.Count -1 do
      with PSourceFileRec(fFileList.Objects[i])^ do
        CopyFile(
          pchar(FullPath),
          pchar(format('src%3.3d.pdf',[i+1])),
          false);
    s := format('"%spdftk.exe" src*.pdf cat output dst.pdf %s %s %s dont_ask',
      [path, allowParams, ownerParam, userParam]);
    executeCommand(s);
  end else
  begin
    actionParams := '';
    fileParams := '';
    for i := 0 to fFileList.Count -1 do
      with PSourceFileRec(fFileList.Objects[i])^ do
      begin
        c := chr(ord('A')+ i);
        s := Format('src%s.pdf', [c]);
        CopyFile(pchar(FullPath), pchar(s), false);
        fileParams := format('%s %s="%s"',[fileParams, c, s]);
        actionParams := actionParams + FmtPagesString(Pages, IsValidPageRanges, c);
        if not IsValidPageRanges then
          raise Exception.CreateFmt(s_invalidPageRange, [i+1]);
      end;
    s := format('"%spdftk.exe" %s cat %s output dst.pdf %s %s %s dont_ask',
      [path, fileParams, actionParams, allowParams, ownerParam, userParam]);
    executeCommand(s);
  end;

  if FileExists(SaveDialog1.FileName) then
    DeleteFile(SaveDialog1.FileName);
  RenameFile('dst.pdf', SaveDialog1.FileName);
end;
//------------------------------------------------------------------------------

procedure TMainForm.DoSplitFile;
var
  i,j: integer;
  s, savedName: string;
  sr: TSearchRec;
begin
  ChDir(tmpPath);
  //circumvent unicode filename issues ...
  CopyFile(pchar(lblSplitSrc.Caption), 'src.pdf', false);

  s := format('%spdftk.exe A="%s" burst output "%s" %s %s %s dont_ask',
    [path, 'src.pdf', 'src_%03d.pdf', allowParams, ownerParam, userParam]);
  executeCommand(s) ;

  j := 0;
  savedName := ChangeFileExt(lblSplitSrc.Caption, '');
  i := FindFirst('src_*.*', faAnyFile, sr);
  while i = 0 do
  begin
    Inc(j);
    SetFileAttributes(pchar(sr.Name), 0); //remove read-only etc
    RenameFile(sr.Name, Format('%s_%3.3d.pdf', [savedName, j]));
    i := FindNext(sr);
  end;
  FindClose(sr);
  SaveDialog1.FileName := lblSplitSrc.Caption;
end;
//------------------------------------------------------------------------------

procedure TMainForm.DoBackgroundStamp(IsBackgroundOp: boolean);
var
  action, s: string;
begin
  if IsBackgroundOp then action := 'background' else action := 'stamp';
  ChDir(tmpPath);
  CopyFile(pchar(lblSource_background.Caption), 'src.pdf', false);
  CopyFile(pchar(lblBackground.Caption), 'src2.pdf', false);

  if cbPageOneOnly.Checked and
    (GetPdfPageCount(lblSource_background.Caption) > 1) then
  begin
    //split off page 1 to tmp1.pdf ...
    s := format('"%spdftk.exe" A=src.pdf cat A1 output tmp1.pdf %s %s %s dont_ask',
      [path, allowParams, ownerParam, userParam]);
    executeCommand(s);
    //create tmp3.pdf from backgrounded page 1 ...
    s := format('%spdftk.exe tmp1.pdf %s src2.pdf output tmp3.pdf %s %s %s dont_ask',
      [path, action, allowParams, ownerParam, userParam]);
    executeCommand(s);

    //split off page 2 to end to tmp2.pdf ...
    s := format('"%spdftk.exe" A=src.pdf cat A2-end output tmp2.pdf %s %s %s dont_ask',
      [path, allowParams, ownerParam, userParam]);
    executeCommand(s);
    //join the result back into one pdf document ...
    s := format('"%spdftk.exe" A=tmp3.pdf B=tmp2.pdf cat A B output dst.pdf %s %s %s dont_ask',
      [path, allowParams, ownerParam, userParam]);
    executeCommand(s);
  end else
  begin
    executeCommand(format('%spdftk.exe src.pdf %s src2.pdf output dst.pdf %s %s %s dont_ask',
      [path, action, allowParams, ownerParam, userParam]));
  end;
  if FileExists(SaveDialog1.FileName) then
    DeleteFile(SaveDialog1.FileName);
  RenameFile('dst.pdf', SaveDialog1.FileName);
end;
//------------------------------------------------------------------------------

procedure TMainForm.DoPageNumber;
var
  i, startPage, pageCount: integer;
  s: string;
  s2, s3: AnsiString;
  rs: TResourceStream;
begin
  ChDir(tmpPath);
  pageCount := GetPdfPageCount(lblSource_background.Caption);
  if cbPageOneOnly.Checked then startPage := 2 else startPage := 1;
  if startPage > pageCount then raise Exception.Create(s_noPagesToNumberError);
  //load blank page PDF from resource that has page number template ...
  rs := TResourceStream.Create(hInstance, 'PAGE', 'TEXT');
  try
    setLength(s2, rs.Size);
    rs.Read(s2[1], rs.Size);

    CopyFile(pchar(lblSource_background.Caption), 'src.pdf', false);
    s := format('"%spdftk.exe" src.pdf burst output src%%03d.pdf dont_ask', [path]);
    executeCommand(s);
    if cbPageOneOnly.Checked then //don't stamp the first page
      RenameFile('src001.pdf', 'dst001.pdf');
    i := startPage;
    while (i <= pageCount) do
    begin
      //create a page numbered pdf...
      s3 := AnsiString(Format(string(s2), [fNumSize, fNumOffset.cx, fNumOffset.cy,
        i, fNumPageSize.cx, fNumPageSize.cy]));
      with TFileStream.Create('1.pdf', fmCreate) do
      try Write(s3[1], length(s3)); finally free; end;
      //stamp each src page with the numbered page ...
      s := format('"%spdftk.exe" src%3.3d.pdf stamp 1.pdf output dst%3.3d.pdf dont_ask',
        [path, i, i]);
      executeCommand(s);
      inc(i);
    end;
  finally
    rs.Free;
  end;
  s := format('"%spdftk.exe" dst*.pdf cat output dst.pdf %s %s %s dont_ask',
    [path, allowParams, ownerParam, userParam]);
  executeCommand(s);
  if FileExists(SaveDialog1.FileName) then
    DeleteFile(SaveDialog1.FileName);
  RenameFile(tmpPath + 'dst.pdf', SaveDialog1.FileName);
end;
//------------------------------------------------------------------------------

procedure TMainForm.DoRotate;
var
  i, j: integer;
  s, action, actionParams: string;
begin
  if rbRotRight.Checked then
    action := 'R'
  else if rbRotLeft.Checked then
    action := 'L'
  else
    action := 'D';

  i := strtoint(eRotPagesFrom.Text);
  j := strtoint(eRotPagesTo.Text);
  if j < i then j := i;

  if i < 0 then
    exit //should never happen
  else if i = 1 then
    actionParams := ''
  else if i = 2 then
    actionParams := '1 '
  else
    actionParams := format('1-%d ', [i-1]);

  if j > i then
    actionParams := actionParams + format('%d-%d%s', [i,j, action]) else
    actionParams := actionParams + format('%d%s', [i,action]);

  if j = fRotatePageCnt then
    //do nothing
  else if j = fRotatePageCnt -1 then
    actionParams := actionParams + format(' %d', [fRotatePageCnt])
  else
    actionParams := actionParams + format(' %d-%d', [j+1, fRotatePageCnt]);

  ChDir(tmpPath);
  CopyFile(pchar(lblSource_rotate.Caption), 'src.pdf', false);

  s := format('"%spdftk.exe" src.pdf cat %s output dst.pdf %s %s %s dont_ask',
    [path, actionParams, allowParams, ownerParam, userParam]);
  executeCommand(s);
  if FileExists(SaveDialog1.FileName) then
    DeleteFile(SaveDialog1.FileName);
  RenameFile(tmpPath + 'dst.pdf', SaveDialog1.FileName);
end;
//------------------------------------------------------------------------------

procedure TMainForm.btnBuildClick(Sender: TObject);
begin
  ErrorNum := 0;
  btnBuild.Enabled := false;
  try try
    if not FileExists(path+'pdftk.exe') then
      raise Exception.Create('pdftk.exe'+ s_doesNotExistInAppFolder);

    if (eUserPassword.text = '') then
      userParam := '' else
      userParam := 'user_pw ' + eUserPassword.text;

    if (eOwnerPassword.text = '') then
      ownerParam := '' else
      ownerParam := 'owner_pw ' + eOwnerPassword.text;

    allowParams := '';
    if (ownerParam <> '') or (userParam <> '') then
    begin
      if cbAllowAll.Checked then
        allowParams := 'allow allfeatures '
      else
      begin
        if cbPrinting.Checked then allowParams := allowParams + 'printing ';
        if cbDegradedPrinting.Checked then allowParams := allowParams + 'degradedprinting ';
        if cbModifyContents.Checked then allowParams := allowParams +'modifycontents ';
        if cbAssembly.Checked then allowParams := allowParams +'assembly ';
        if cbCopyContents.Checked then allowParams := allowParams +'copycontents ';
        if cbScreenReaders.Checked then allowParams := allowParams +'screenreaders ';
        if cbModifyAnnotations.Checked then allowParams := allowParams +'modifyannotations ';
        if cbFillIn.Checked then allowParams := allowParams +'fillin ';
        if allowParams <> '' then allowParams := 'allow ' + allowParams;
      end;
    end;

    DeleteAllTmpFiles;
    ForceDirectories(tmpPath);

    if PageControl1.ActivePage = tsCollate then
    begin
      if (fFileList.Count = 0) or not SaveDialog1.Execute then exit;
      DoCollateFiles;                                //JOIN
    end
    else if PageControl1.ActivePage = tsSplit then
      DoSplitFile                                   //SPLIT
    else if PageControl1.ActivePage = tsBackground then
    begin
      if not fileExists(lblSource_background.Caption) or
        not SaveDialog1.Execute then exit;

      if rbPageNumber.Checked then
        DoPageNumber else                           //PAGE NUMBER
        DoBackgroundStamp(rbBackground.Checked);    //BACKGROUND/STAMP
    end
    else if PageControl1.ActivePage = tsRotate then
    begin
      if not SaveDialog1.Execute then exit;
      DoRotate;                                     //ROTATE
    end;
    statusbar1.SimpleText :=
      '  ' + ExtractFilename(SaveDialog1.FileName) + ' ' + s_saved;
  except
    on E: Exception do
      ProcessError(E.Message);
  end;
  finally
    btnBuild.Enabled := true;
    DeleteAllTmpFiles;
  end;

end;
//------------------------------------------------------------------------------

procedure TMainForm.cbAllowAllClick(Sender: TObject);
begin
  if not cbAllowAll.Focused then exit;
  cbPrinting.Checked := cbAllowAll.Checked;
  cbDegradedPrinting.Checked := cbAllowAll.Checked;
  cbModifyContents.Checked := cbAllowAll.Checked;
  cbAssembly.Checked := cbAllowAll.Checked;
  cbCopyContents.Checked := cbAllowAll.Checked;
  cbScreenReaders.Checked := cbAllowAll.Checked;
  cbModifyAnnotations.Checked := cbAllowAll.Checked;
  cbFillIn.Checked := cbAllowAll.Checked;
end;
//------------------------------------------------------------------------------

procedure TMainForm.cbPrintingClick(Sender: TObject);
begin
  //if any checkbox is unticked then make sure 'all actions' is unticked too.
  if (Sender is TCheckbox) and
    TCheckbox(Sender).Focused and not TCheckbox(Sender).Checked then
      cbAllowAll.Checked := false;
end;
//------------------------------------------------------------------------------

procedure TMainForm.btnSrcBackgroundClick(Sender: TObject);
begin
  OpenDialog1.Options := OpenDialog1.Options - [ofAllowMultiSelect];
  if not OpenDialog1.Execute then exit;
  lblSource_background.Caption := OpenDialog1.FileName;
  UpdateButtonsAndMenus(0);
end;
//------------------------------------------------------------------------------

procedure TMainForm.btnBackgroundClick(Sender: TObject);
begin
  OpenDialog1.Options := OpenDialog1.Options - [ofAllowMultiSelect];
  if not OpenDialog1.Execute then exit;
  lblBackground.Caption := OpenDialog1.FileName;
  UpdateButtonsAndMenus(0);
end;
//------------------------------------------------------------------------------

procedure TMainForm.About1Click(Sender: TObject);
begin
  with TAboutForm.Create(self) do
  try
    showModal;
  finally
    free;
  end;
end;
//------------------------------------------------------------------------------

procedure TMainForm.Contents1Click(Sender: TObject);
begin
  ShellExecute(0, nil, 'http://www.angusj.com/pdftkb/', Nil, Nil, SW_NORMAL);
end;
//------------------------------------------------------------------------------

procedure TMainForm.btnSplitSrcClick(Sender: TObject);
begin
  OpenDialog1.Options := OpenDialog1.Options - [ofAllowMultiSelect];
  if not OpenDialog1.Execute then exit;
  lblSplitSrc.Caption := OpenDialog1.FileName;
  UpdateButtonsAndMenus(0);
end;
//------------------------------------------------------------------------------

procedure TMainForm.eRotPagesFromChange(Sender: TObject);
var
  val: integer;
  edit: TEdit;
  OkVal: boolean;
begin
  edit := TEdit(Sender);
  val := StrToIntDef(edit.Text, -1);
  OkVal := (val > 0) and (val <= fRotatePageCnt);
  if OkVal then
    edit.Font.Color := clBlack else
    edit.Font.Color := clRed;
  btnBuild.Enabled := (lblSource_rotate.Caption <> '') and
    (eRotPagesFrom.Font.Color = clBlack) and (eRotPagesTo.Font.Color = clBlack);
end;
//------------------------------------------------------------------------------

procedure TMainForm.btnSrcRotateClick(Sender: TObject);
begin
  if not OpenDialog1.Execute then exit;
  lblSource_rotate.Caption := OpenDialog1.FileName;
  fRotatePageCnt := GetPdfPageCount(OpenDialog1.FileName);
  eRotPagesFrom.Text := '1';
  eRotPagesTo.Text := IntToStr(fRotatePageCnt);
  UpdateButtonsAndMenus(0);
end;
//------------------------------------------------------------------------------

procedure TMainForm.tsCollateResize(Sender: TObject);
var
  i,j, w: integer;
begin
  w := StringGrid1.ClientWidth;
  with StringGrid1 do
    for i := 0 to RowCount -1 do
    begin
      j := canvas.TextWidth(Cells[0,i])+4;
      if j > w then w := j;
    end;
  StringGrid1.ColWidths[0] := w;
end;
//------------------------------------------------------------------------------

procedure TMainForm.LoadIniFileInfo;
var
  fn: string;
begin

  fn := ChangeFileExt(paramstr(0),'.ini');
  with TIniFile.Create(fn) do
  try
    //position form where last opened ...
    origin.X := ReadInteger(user, 'PosLeft', 0);
    origin.Y := ReadInteger(user, 'PosTop', 0);
    Expanded := ReadBool(user, 'Expanded', false);

    lblBackground.Caption := ReadString(user, 'BackgroundFile','');
    if (lblBackground.Caption <> '') and not FileExists(lblBackground.Caption) then
      lblBackground.Caption := '';

    //page numbering params with dimensions in 'points'(default size = A4)
    //see also http://www.printernational.org/iso-paper-sizes.php
    fNumSize := ReadInteger('Numbering', 'NumSize', 10); //ie font size
    fNumPageSize.cx := ReadInteger('Numbering', 'PageWidth', 595);
    fNumPageSize.cy := ReadInteger('Numbering', 'PageHeight', 842);
    fNumOffset.cx := ReadInteger('Numbering', 'NumOffsetX', 560);
    fNumOffset.cy := ReadInteger('Numbering', 'NumOffsetY', 30);
  finally
    free;
  end;
end;
//------------------------------------------------------------------------------

procedure TMainForm.SaveIniFileInfo;
var
  fn: string;
begin
  fn := ChangeFileExt(paramstr(0),'.ini');
  with TIniFile.Create(fn) do
  try
    WriteInteger(user, 'PosLeft', left);
    WriteInteger(user, 'PosTop', top);
    WriteString(user, 'BackgroundFile', lblBackground.Caption);
    WriteBool(user, 'Expanded', expanded);

    WriteInteger('Numbering', 'PageWidth', fNumPageSize.cx);
    WriteInteger('Numbering', 'PageHeight', fNumPageSize.cy);
    WriteInteger('Numbering', 'NumOffsetX', fNumOffset.cx);
    WriteInteger('Numbering', 'NumOffsetY', fNumOffset.cy);
    WriteInteger('Numbering', 'NumSize', fNumSize);
  finally
    free;
  end;
end;
//------------------------------------------------------------------------------

procedure TMainForm.FormCloseQuery(Sender: TObject; var CanClose: Boolean);
begin
  SaveIniFileInfo;
end;
//------------------------------------------------------------------------------

procedure TMainForm.rbPageNumberClick(Sender: TObject);
begin
  if rbPageNumber.Checked then
  begin
    btnBackground.Enabled := false;
    btnBackground.Caption := '';
    lblBackground.Visible := false;
    cbPageOneOnly.Caption := s_skipFirstPageNumbering;
  end else if rbBackground.Checked then
  begin
    btnBackground.Enabled := true;
    btnBackground.Caption := s_chooseBackground;
    lblBackground.Visible := true;
    cbPageOneOnly.Caption := s_backgroundFirstPageOnly;
  end else
  begin
    btnBackground.Enabled := true;
    btnBackground.Caption := s_chooseStamp;
    lblBackground.Visible := true;
    cbPageOneOnly.Caption := s_stampFirstPageOnly;
  end;
  UpdateButtonsAndMenus(0);
end;
//------------------------------------------------------------------------------

procedure TMainForm.RotatePages1Click(Sender: TObject);
begin
  PageControl1.ActivePageIndex := TMenuItem(Sender).GroupIndex;
end;
//------------------------------------------------------------------------------


{$R-}
procedure MakeMulTable;
var
  I, J: Integer;
const
  Div255 = 1 / 255;
begin
  for I := 0 to 255 do
    MulTable[I, 0] := 0;
  for I := 1 to 255 do
    for J := 0 to 255 do
      MulTable[I, J] := Round(I * J * Div255);
end;
//------------------------------------------------------------------------------
{$R+}

//Below is a workaround for a Windows bug ...
//when opening an Open/Save dialog more than once and hovering over a filename
//causing a hint window to popup, the application will crash on Windows XP Sp2.
//(If you're sceptical about this being a Windows bug then check it out
//in Notepad.exe.)

initialization
  OleInitialize(nil);
  MakeMulTable;

finalization
  OleUninitialize;

end.

program PdftkBuilder;

uses
  Forms,
  main in 'main.pas' {MainForm},
  about in 'about.pas' {AboutForm};

{$R *.res}

begin
  Application.Initialize;
  Application.Title := 'PDFTK Builder';
  Application.HelpFile := 'PdftkBuilder.hlp';
  Application.CreateForm(TMainForm, MainForm);
  Application.Run;
end.

